
exampleLang

exampleLang is a MAL language intended to demonstrate the Maven
project structure of a MAL language.

This project has the following structure:


	The file pom.xml is the Maven configuration file of the project.


	The directory src/main/mal contains the MAL specification
exampleLang.mal, which is the MAL specification of exampleLang.


	The directory src/main/resources/icons contains SVG icons for the
assets in exampleLang.


	The directory src/test/java/org/mal_lang/examplelang/test
contains the unit tests of exampleLang.





Apache Maven

Apache Maven [https://maven.apache.org/] is a build tool and
dependency management tool for Java projects. You can read more about
Maven at https://en.wikipedia.org/wiki/Apache_Maven. Follow the
instructions at https://maven.apache.org/download.cgi to download
Maven, and follow the instructions at
https://maven.apache.org/install.html to install Maven.




Building exampleLang and running the unit tests

The
MAL compiler [https://github.com/meta-attack-language/malcompiler]
compiles MAL specifications (.mal files) into different formats,
using different backends. The reference backend generates Java code
that is suitable for testing purposes and evaluating your language.
The securiCAD backend generates a .jar file that can be used with
foreseeti [https://www.foreseeti.com/]’s products, including
securiCAD [https://www.foreseeti.com/securicad/], which is a tool
that can be used to graphically create models using your language and
to simulate attacks on those models.


Building with the reference backend and running the unit tests

To compile exampleLang with the reference backend of the MAL compiler
and then run the unit tests, execute the following command from the
exampleLang directory:

mvn test





This will invoke the MAL compiler’s reference backend to generate
.java files under target/generated-test-sources. These .java
files and the unit tests in src/test/java will then be compiled
into .class files under target/test-classes. The unit tests will
then finally be executed.

To only compile exampleLang into .java files, execute the following
command:

mvn generate-test-sources





To compile exampleLang into .java files and then compile these
.java files and the unit tests in src/test/java into .class
files, execute the following command:

mvn test-compile





To run a specific test class, execute the following command:

mvn test -Dtest=TestExampleLang





Where TestExampleLang is the test class.

To run a specific test method in a test class, execute the following
command:

mvn test -Dtest=TestExampleLang#testNoPassword





Where TestExampleLang is the test class and testNoPassword is the
test method.




Building a securiCAD compatible .jar file

To build a securiCAD compatible .jar file, you need access to
foreseeti’s maven repository. To request access, please contact
support@foreseeti.com. When you have received your credentials, you
can store them in a file ~/.aws/credentials
(%UserProfile%\.aws\credentials on windows). For example:

[default]
aws_access_key_id=AKIAIOSFODNN7EXAMPLE
aws_secret_access_key=wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY





To compile exampleLang with the securiCAD backend of the MAL
compiler, execute the following command:

mvn package -PsecuriCAD





The resulting .jar file will be located in
target/examplelang-1.0.0.jar.

If you don’t want to run the unit tests when building a securiCAD
compatible .jar file, execute the following command:

mvn clean package -PsecuriCAD -Dmaven.test.skip=true










Using exampleLang as a template MAL language

To create a new language using exampleLang as a template, you need to
do the following:


	Create a new MAL language project using exampleLang as the template


	cp -r exampleLang/ myLang/






	Enter the directory of the new MAL language project


	cd myLang/






	Remove build scripts


	rm -rf .buildscript


	rm .travis.yml






	Update LICENSE with a license of your choice


	Update copyright notices to reflect your license in


	NOTICE


	pom.xml


	src/main/mal/exampleLang.mal


	src/test/java/org/mal_lang/examplelang/test/*.java










	Update README.md with relevant information about your language.
Information about how to use Markdown can be found at
https://help.github.com/en/articles/basic-writing-and-formatting-syntax.


	Update pom.xml to reflect your project


	Update <groupId> with a reverse domain name that you can use


	Example: com.example






	Update <artifactId> with a suitable name


	Example: mylang






	Update <version> with the version of your language


	Example: 1.0.0






	Update <name> with the name of your language


	Example: myLang






	Update <mal.file> with the name of the main MAL specification
of your language


	Example: myLang.mal






	Update <mal.securicad.package> with the package name of your
language


	Example: com.example.mylang






	Update <mal.reference.package> with the test package name of
your language


	Example: com.example.mylang.test










	Rename src/main/mal/exampleLang.mal to the name of the main MAL
specification of your language


	mv src/main/mal/exampleLang.mal src/main/mal/myLang.mal






	Update your main MAL specification’s #id and #version


	Example: #id: "com.example.mylang", #version: "1.0.0"






	Rename unit tests in src/test/java to reflect your language


	Change the package name of the unit tests to the test package name
of your language


	Example: package com.example.mylang.test;











License

Copyright © 2020 Foreseeti AB [https://www.foreseeti.com/]

All files distributed in the exampleLang project are licensed under the Apache License, Version 2.0 [https://www.apache.org/licenses/LICENSE-2.0], except for the following files:

| File | License |
| — | — |
| Host.svg | [image: Host.svg] “Computer [https://thenounproject.com/term/computer/576625/]” icon by ✦ Shmidt Sergey ✦ [https://thenounproject.com/monstercritic/] from the Noun Project [https://thenounproject.com/] is licensed under CC BY 3.0 [https://creativecommons.org/licenses/by/3.0/]. |
| Network.svg | [image: Network.svg] “Network [https://thenounproject.com/term/network/691907/]” icon by ✦ Shmidt Sergey ✦ [https://thenounproject.com/monstercritic/] from the Noun Project [https://thenounproject.com/] is licensed under CC BY 3.0 [https://creativecommons.org/licenses/by/3.0/]. |
| Password.svg | [image: Password.svg] “Lock [https://thenounproject.com/term/lock/576530/]” icon by ✦ Shmidt Sergey ✦ [https://thenounproject.com/monstercritic/] from the Noun Project [https://thenounproject.com/] is licensed under CC BY 3.0 [https://creativecommons.org/licenses/by/3.0/]. |
| User.svg | [image: User.svg] “User [https://thenounproject.com/term/user/581261/]” icon by ✦ Shmidt Sergey ✦ [https://thenounproject.com/monstercritic/] from the Noun Project [https://thenounproject.com/] is licensed under CC BY 3.0 [https://creativecommons.org/licenses/by/3.0/]. |

See LICENSE and NOTICE for details.

  IT systems are growing in complexity, and due to digitization and technological advancements, cyber attacks are increasing.  Parts of our world, such as cars,power grids and homes are now connected to the Internet.

In order to keep these systems secure, the first thing is to understand and map out different threatsand attacks on them.  Threat modeling is a process that can be used to analyze potential attacks or threats, in order to make sure the systems are designed bysecurity.

This wiki is the description and hands-on manual of the Meta Attack Lan-guage (MAL), which is a threat modeling language framework for the creation of domain specific languages (DSL). In brief, MAL combines probabilistic attack defense graphs with object oriented modeling.  By using MAL it is possible to automate the security analysis of modeled systems. MAL is developed at KTH Royal Institute of Technology [https://www.kth.se/nse/research/software-systems-architecture-and-security/projects/mal-the-meta-attack-language-1.922174]. Several  domain  specific  languages  have  been  built  in  MAL  serving  as  examples  of  how  a  language  can  be  built. In addition to the research prototypes developed at KTH, the company foreseeti [https://www.foreseeti.com/] has developed a commercial product supporting language development and model simulations in MAL.



            

          

      

      

    

  

    
      
          
            
  
Versioning

This page documents instructions for the legacy compiler. The documentation for the current foreseeti compiler can be found here [https://github.com/mal-lang/mal-documentation/wiki/Compiling].



The compiler generates Java code from the MAL specification. More details regarding compilation can be found in the MAL Compiler User Guide and in the GIT repository [https://github.com/mal-lang]. MAL is compiled with the following commands:

| Assertion method | Description
| — | — |
| assertUncompromised() | Check for unsuccessful compromise of the attack step.
| assertUncompromisedFrom(parent) | Check for unsuccessful compromise of the attack step from the specified parent attack step.
| assertCompromisedInstantaneously() | Check for successful and immediate compromise of the attack step.
| assertCompromisedWithEffort() | Check for successful compromise of the attack step but only after some effort/time is spent.
| assertCompromisedInstantaneouslyFrom() | Same as assertCompromisedInstantaneously from specified parent attack step.
| assertCompromisedWithEffortFrom() | Same as assertCompromisedWithEffort from specified parent attack step.

java -cp mal-compiler-0.0.X.jar se.kth.mal.Master -i <input_path>  
-o <ouput_path> -p <package_name> -t <tests_path>


usage: 
 X to be replaced by the version number. 
 -f,--foreseeti       flag to use foreseeti backend
 -i,--input <arg>     input mal file path
 -o,--output <arg>    output folder path for generated code
 -p,--package <arg>   package name of generated code
 -t,--tests <arg>     output folder path for generated test code
 -v,--visual <arg>    icons for visualization





The output of a successful compilation will look like:

>>> Using MAL Compiler <<<
Reading from <input>.malinc
...
Attempting to write the TTC Config File...
Complete





Several files will be created in the specified output folder. One of these is the html file, which can be opened in a web browser to visually represent the attack graph. The attack graph’s large circles are the assets, the small circles are the attack steps and the squares are defences.





            

          

      

      

    

  

    
      
          
            
  
Installation

Instructions regarding the MAL compiler and the MAL project structure can be found in the exampleLang README [https://github.com/mal-lang/exampleLang/blob/master/README].




Compilation primer

For detailed instructions, see building exampleLang and running unit tests (exampleLang) [https://github.com/mal-lang/exampleLang/blob/master/README.md#building-examplelang-and-running-the-unit-tests].

For instructions regarding how to write tests, see Instantiating Language Models [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models].


Compile and run unit tests

mvn test





Generates .java files from the MAL specifications and outputs them into <project>/target/generated-test-sources. Also compiles the language source files together with any unit test code found. Finally runs the unit tests.




Compile and run a specific test file

mvn test -Dtest=TestMyLanguage





Same as mvn test except only the test cases defined in TestMyLanguage.java are run.




Compile and run a specific test case

mvn test -Dtest=TestMyLanguage#myTestCase





Same as mvn test except only the test case myTestCase defined in TestMyLanguage.java is run.




Build for securiCAD

mvn clean package -PsecuriCAD





The result will be a .jar file found in <project>/target. To skip running tests, add -Dmaven.test.skip=true.

Building a securiCAD package requires access to foreseeti’s Maven repository. Contact support@foreseeti.com to request access.




Generate Java source

mvn generate-test-sources





This solely compiles MAL specifications into .java source code files. The files are found under <project>/target/generated-test-sources.




Generate and compile Java source

mvn test-compile





This compiles MAL specifications into .java source code files. Also compiles those into .class files together with any unit test files found. The Java source files are found under <project>/target/generated-test-sources and the class files under <project>/target/test-classes. No tests are run with this command.







            

          

      

      

    

  

    
      
          
            
  For instruction on how to run tests, see Compiling [https://github.com/mal-lang/mal-documentation/wiki/Compiling].

Overview

After the MAL specifications are created, it is time to instantiate models and verify the domain-specific language. A model is an instantiation of some assets and the connections between them. Each model created is considered one test case. In the case of MAL, models are instantiated using Java and tests are executed with JUnit. Test case files must, therefore, be structured accordingly [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#complete-test-file-skeleton]. Additionally, test case files must be placed under the <project>/src/test/java/<Defined>/<Project>/<ID> directory in accordance with the Apache Maven project structure [https://github.com/mal-lang/exampleLang#examplelang]. Finally, test case filenames must always begin with Test, e.g., TestExampleLang.java.

MAL test cases [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#complete-test-case-example] conventionally follow a 4-part structure:


	Instantiate assets as Java objects [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#instantiating-assets].


	Define associations [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#define-associations].


	Specify the attacker’s starting point(s) [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#specify-attacker-starting-points].


	Run the simulation and test assertions [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#run-the-simulation-and-test-assertions].




There are two distinct styles of test case definitions, waterfall [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#waterfall] and model-based [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#model-based]. The waterfall style follows the 4-part structure for every test case. They are easy to write and follow, making it good for beginners. The waterfall style, however, has a lot of redundancy and takes longer to write. The model-based style is more streamlined but potentially more difficult to follow when starting out.




The Test File


Setting Up


Complete Test File Skeleton

package <SameAsLanguageID>.test
import core.Attacker;
import core.Asset;
import core.AttackStep;
import core.Defense;
import core.AttackStepMax;
import core.AttackStepMin;

import org.junit.jupiter.api.Test;
import org.junit.jupiter.api.AfterEach;
import org.junit.jupiter.api.DisplayName;

public class Test<SameAsFileName> {

    @Test
    @DisplayName("An example test case")
    public void testExampleCase() {
        //Test code here...
    }

    @AfterEach
    public void deleteModel() {
        // Clean up before each test
        Asset.allAssets.clear();
        AttackStep.allAttackSteps.clear();
        Defense.allDefenses.clear();
    }
}








Dissecting the Test File Skeleton


Package Definition

// Generic syntax
//package <SameAsLanguageID>.test;

// Example
package com.lang.example.test;





The first line in each file should be the package definition. This is the same as the project ID for the MAL project excecpt .test is appended.




Import Statements

import core.*;





Every test file typically begins by importing fundamental MAL components. These are Attacker, AttackStep, Asset, Defense, AttackStepMin, and AttackStepMax. Importing all of them is not necessary. Only import the ones actually used.

import org.junit.jupiter.api.Test;
import org.junit.jupiter.api.AfterEach;
import org.junit.jupiter.api.DisplayName;





The appropriate JUnit components must be imported to run tests. The Test component is necessary for defining tests. AfterEach is typically used to remove lingering objects before each test case. DisplayName allows for setting custom display names, commonly shown in IDEs. Any JUnit components may be imported but the three mentioned are the most common.




Test Class Declaration

// Generic syntax
// public class Test<SameAsFileName> {...}

public class TestExampleLang {...}





Defines the class containing test case definitions. The class name must be identical to the file name, excluding the file extension.




Test Case Definition

@Test
@DisplayName("An example test case")
public void testExampleCase() {
    //Test code here...
}





@Test designates the function as a test case. @DisplayName sets a custom name for the case. JUnit test cases must always be declared as public functions without return types (void). The test code is regular java code following the 4-part MAL test case structure. Test cases can be defined in any order since JUnit randomizes the order each time tests are run.

@AfterEach
public void deleteModel() {
    Asset.allAssets.clear();
    AttackStep.allAttackSteps.clear();
    Defense.allDefenses.clear();
}





This function is executed after each test case. The code above clears out all lingering assets between each test.








Writing Test Cases


Complete Test Case Example

@Test
@DisplayName("An example test case")
public void testExampleCase() {
    // Create assets
    Computer computer = new Computer("Linux", false); // Disable some defense
    // Computer computer = new Computer("Linux", true); // Enable some defense
    User alice = new User("Alice");

    // Create associations
    Computer.addUsers(alice);

    // Create attacker
    Attacker attacker = new Attacker();
    attacker.addAttackPoint(computer.attemptToHack);

    // Run simulation and test assertions
    attacker.attack();

    computer.compromised.assertCompromisedInstantaneously();
    alice.hacked.assertCompromisedInstantaneously();
}








Dissecting the Test Case Example


Instantiating Assets

// Generic syntax
// MyAsset assetName = new MyAsset("NameForTestPrintOuts", defense1, [...], defensen);

// Example
Computer computer = new Computer("Linux", false);
// Computer computer = new Computer("Linux", true);





Each non-abstract Asset defined with MAL has its own Java class definition. They are instantiated like regular Java objects. The string NameForTestPrintOuts represents the object instance in test log printouts.




Define Associations

Computer computer = new Computer("Linux");
User alice = new User("Alice");

// Generic syntax
// asset1.add<Role>(asset2);


// Example
Computer.addUsers(alice);
// alice.addComputers(computer); // Same association, reverse direction.





Every asset has an automatically generated add function for each association. The function names are determined by the roles [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#association] defined in the MAL specifications. Add functions always have the form add<Role>, where the first letter of the role name is always capitalized.




Specify Attacker Starting Point(s)

Attacker attacker = new Attacker();
attacker.addAttackPoint(computer.attemptToHack);





One attacker object must be instantiated for every test case. The addAttackPoint function is used to specify which attack steps are compromised upon starting the simulation. In this example, attemptToHack is an attack step for an instantiated Computer asset.




Run the Simulation and Test Assertions

Attacker attacker = new Attacker();
attacker.addAttackPoint(computer.attemptToHack);
attacker.attack();

computer.hacked.assertCompromisedInstantaneously();
alice.hacked.assertCompromisedInstantaneously();





Calling attack() for the Attacker object without parameters starts the simulation. Nothing will happen unless this is called. The last step is to write assertion clauses [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#assertion-methods]. These check whether the specified attack steps were reachable or unreachable. The test succeeds if all results conformed to the assertions. The test fails if at least one contradiction is produced. In this example, the attacker starts by attempting to hack a computer. The assertion checks if the corresponding compromised step was reached as expected. Then, it checks whether the user, Alice, was hacked as a result.








Assertion Methods

| Assertion method | Description
| — | — |
| assertUncompromised() | Check for unsuccessful compromise of the attack step.
| assertUncompromisedFrom(parent) | Check for unsuccessful compromise of the attack step from the specified parent attack step.
| assertCompromisedInstantaneously() | Check for successful and immediate compromise of the attack step.
| assertCompromisedWithEffort() | Check for successful compromise of the attack step but only after some effort/time is spent.
| assertCompromisedInstantaneouslyFrom() | Same as assertCompromisedInstantaneously from specified parent attack step.
| assertCompromisedWithEffortFrom() | Same as assertCompromisedWithEffort from specified parent attack step.




Test Case Styles


Waterfall

The waterfall style is when the 4-step structure is repeated for every test case. It is also the same one demonstrated on this page. Waterfall test cases are easy to write and follow. However, they tend to be lengthy and contain a lot of redundancy. The waterfall style is, therefore, only recommended for practicing the basics. Other styles, such as the model-based style [https://github.com/mal-lang/mal-documentation/wiki/Instantiating-Language-Models#model-based], should be used eventually.


Waterfall Example

@Test
@DisplayName("A waterfall example")
public void testExampleCase() {
    // Create assets
    Computer computer = new Computer("Linux", false);
    User alice = new User("Alice");

    // Create associations
    Computer.addUsers(alice);

    // Create attacker
    Attacker attacker = new Attacker();
    attacker.addAttackPoint(computer.attemptToHack);

    // Run simulation and test assertions
    attacker.attack();

    computer.compromised.assertCompromisedInstantaneously();
    alice.hacked.assertCompromisedInstantaneously();
}










Model-based

The model-based, or object-oriented, style creates additional private classes within the test class. These are referred to as models. Models can, for example, instantiate and associate any number of pre-defined assets according to parameters sent to constructors. This style is very efficient when many test cases require similar, and possibly complex, setups. The drawback with this style is that test cases can become more difficult to follow, especially when handed to other developers.


Model-based Example

private class ComputerModel{
    // Basic assets + associations model
    public final User alice = new User("Alice");
    public Computer computer;

    public ComputerModel(boolean isDefended){
        computer = new Computer("Linux", isDefended)
        computer.addUsers(alice);
    }
}

@Test
@DisplayName("A model-based example")
public void testExampleCase() {
    // Create model (assets + associations)
    ComputerModel model = new ComputerModel(false);

    // Create attacker
    Attacker attacker = new Attacker();
    attacker.addAttackPoint(model.computer.attemptToHack);

    // Run simulation and test assertions (models can also include attackers and assertions!)
    attacker.attack();

    model.computer.compromised.assertCompromisedInstantaneously();
    model.alice.hacked.assertCompromisedInstantaneously();
}















            

          

      

      

    

  

    
      
          
            
  This page contains MAL source code examples. For a syntax overview, see MAL Syntax [https://github.com/mal-lang/mal-documentation/wiki/MAL-Syntax#MAL-Symbols].

– Existence (E) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#existence-e]

– Non-existence (!E) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#non-existence-e]

– Require (<-) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#require--]

– TypeOf ([]) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#typeof-]

– Union (\/) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#union]

– Intersection (/\) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#intersection-]

– Difference (-) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#difference-]

– Collect operator (X.A) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#collect-operator-xa]

– Transitive operator (X*.A) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#transitive-operator-xa]

– Let substitution (X().A) [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#let-substitution-xa]

– Hidden step [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#hidden-step]

– Debug step [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#-debug-step]

– Trace origin [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#trace-origin]

– Risk type [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#risk-type]




Basic MAL Example

MAL, like other languages, has its own syntax and alphabet of symbols. Each asset may have attack or defense steps, and can have associations between each other. The MAL symbols can be seen below. The associations between assets follow the structure of UML class diagrams, where associations are bound to types with cardinalities.

category System {
    asset Computer {
        let allFolders = folder.subFolder*        

        | connect
            -> access
        E firewallExists
            <- firewall
            -> firewall.bypass
        E! noFirewall
            <- firewall
            -> firewallBypassed
        | firewallBypassed @hidden
            -> access
        | vulnerability
            -> compromise
        & access
            -> compromise
	& compromise
            ->  allFolders().accessFolder
                // Let substitution
    }
    asset Folder {
        | accessFolder
            -> stealSecrets
        | stealSecrets
    }
}

category Security {
    asset Firewall {
        & bypass [Bernoulli(0.2)]
            -> computer.firewallBypassed
        # hardened
            -> bypass
    }
}

associations {
    Computer [computers] * <-- Protect --> 0..1 [firewall] Firewall
    Computer [computer] * <-- Contains --> * [folder] Folder
    Folder [folder] 1 <-- Contains --> * [subFolder] Folder
}








MAL Demonstrations


Language Constituents


Define

// Generic syntax
#key1: "value"

// Intuition
The ID "key1" contains the project-wide information "value"

// Example
#id: "com.code.example.MAL"
#version: "1.0.0"

//Interpretation
The name of the MAL project is MAL.example.code.com (reverse domain name notation). The current version of the language is 1.0.0.








Include

// Simplified syntax
include "./my/spec.mal"

// Intuition
Include the source code of spec.mal in the current specification.

// Example
include "networking.mal"
include "systems/linux.mal

//Interpretation
Includes the source code from networking into the current specification. Additionally includes the source code in linux.mal found under the systems directory relative to the current specification. It is now possible to reference assets and attack steps of both networking.mal and linux.mal.








Category

// Simplified syntax
category foo
{
[...]
}

// Intuition
Assets defined within the curly braces are categorized as foo.

// Example
category System
{
[...]
}

category Security
{
[...]
}

//Interpretation
Asset defined within the first category are classified as System assets. Assets defined within the second are classified as Security assets.








Asset

// Simplified syntax
asset Foo
{
[...]
}

// Intuition
The MAL representation of a physical or logical object Foo.

// Example
asset Computer
{
[...]
}

asset Software
{
[...]
}

//Interpretation
Computer is the MAL representation of a physical computer system. Software is the MAL representation of executable applications.








Extends

// Simplified syntax
asset Parent
{
[parent logic]
}

asset Child extends Parent
{
[parent + child logic]
}

// Intuition
The asset Child is a sub-class of Parent. Child additionally inherits the logic specified for its parent.

// Example
asset OperatingSystem
{
[OperatingSystem logic]
}

asset Linux
{
[OperatingSystem + Linux logic]
}

//Interpretation
Linux is a type of OperatingSystem. Therefore, what applies for generic OperatingSystems also applies to all instances of Linux systems in the language.








Abstract asset

// Simplified syntax
abstract asset Foo
{
[Foo logic]
}

asset Bar extends Foo
{
[Foo + Bar logic]
}

// Intuition
The MAL representation of the object Foo, which cannot be instantiated. However, its sub-class Bar may be instantiated.

// Example
abstract asset OperatingSystem
{
[OperatingSystem logic]
}

asset Linux
{
[OperatingSystem + Linux logic]
}

//Interpretation
The language contains logic for generic operating systems. However, only concrete sub-classes, like Linux, may be instantiated.








User info

// Simplified syntax
asset Foo
  user info: "Hello"
{
  | bar
    user info: "World"
}

// Intuition
"Hello" is information for end users regarding Foo. "World" is end user information regarding the attack step bar in asset Foo.

// Example
asset Software
  user info: Software represents application running on your computer systems.
{
  | attack
    user info: Software can be attacked in multiple ways, which leads to undesirable consequences.
}

//Interpretation
Asset level user information can provide a description of what the asset corresponds to in reality. Logic level user information can explain what the step means.








Developer info

// Simplified syntax
asset Foo
  developer info: "Hello"
{
  | bar
    developer info: "World"
}

// Intuition
"Hello" is information for other MAL developers regarding Foo. "World" is developer information regarding the attack step bar in asset Foo.

// Example
asset Software
  developer info: We do not model specific software at this point in time. Instantiate as needed.
{
  | attack
    developer info: Stub logic until a vulnerability specification is developed.
}

//Interpretation
Asset level developer information can, for example, explain why an asset is needed or if other assumptions have been made. Logic level developer information can, for example, explain why a step is necessary when it is not apparent.








Modeler info

// Simplified syntax
asset Foo
  modeler info: "Hello"
{
  | bar
    modeler info: "World"
}

// Intuition
"Hello" is information for parser developers or modelers regarding Foo. "World" is parser/modeler information regarding the attack step bar in asset Foo.

// Example
asset Software
  developer info: Give appropriate names during parsing.
{
  | attack
    developer info: Reached from associated vulnerabilities. Attach to Software during parsing based on vulnerability scanner output.
}

//Interpretation
Asset level modeler information can, for example, explain how the asset should be treated during the parsing process. Logic level developer information can, for example, explain assumptions made by the MAL developers regarding parsing contra language responsibilities.








Let

// Simplified syntax
let name  = foo \/ bar

// Intuition
The union of all assets foo and bar is given the label name.

// Example
asset Computer
{
  let components = software \/ hardware
}

//Interpretation
The name components can be used to refer to the union of software and hardware assets associated with any given instance of Computer.








Association

// Simplified syntax
Asset1 [foo] * <-- connects --> * [bar] Asset2

// Intuition
Any number of Asset1 instantiations can be connected to any number of Asset2 instantiations. Inline references from Asset1 to Asset2 use the name bar. Conversely, Asset2 refers to Asset1 with the name foo.

// Example
Computer [host] 1 <-- runs --> 1-* [os] OperatingSystem

//Interpretation
One Computer must be associated with at least one type of operating system. References from Computer assets to its associated OperatingSystem assets use the format 'os.attackStep'. Reverse references from OperatingSystem assets to the underling Computer use the format 'host.attackStep'.










MAL Symbols


Leads to (->)

// Simplified syntax
| step1
   -> step2,
      step3

// Intuition
If step 1 is reached, then, this leads to the subsequent steps 2 and 3.

// Example
| authenticate
  -> readData,
     writeData,
     deleteData

//Interpretation
If the attacker gains authenticated access, they can now attempt to read, write, and delete data as a result.








Append (+>)

// Simplified syntax
asset Parent
{
  | step1
    -> foo
}

asset Child extends Parent
{
  | step1
    +> [foo,] // 'foo,' is implicit.
       bar
}
// Intuition
The element step1 is defined multiple times in the inheritance hierarchy. Rather than overriding the Parent logic, add any new logic from Child to the pre-existing logic.

// Example
asset OperatingSystem
{
  | spyware
    -> logKeystrokes
}

asset Linux
{
  | spyware
    +> readBashHistory
}

//Interpretation
If a system is infected with spyware, the user's keystrokes may be logged. Additionally, it is assumed that spyware might both log keystrokes and read the Bash history file on Linux systems.








OR-step (|)

// Simplified syntax
| step1
  -> step3
| step2
  -> step3
| step3

// Intuition
Reaching either step1 or step2 is enough to reach step3.

// Example
| guessPassword
  -> authenticate
| stealPassword
  -> authenticate
| authenticate

//Interpretation
An attacker may either guess or steal a password to obtain authenticated access to some asset.








AND-step (&)

// Simplified syntax
| step1
  -> step3
| step2
  -> step3
& step3

// Intuition
Reaching both step1 or step2 is required to reach step3.

// Example
| obtainPassword
  -> authenticate
| obtainMFAToken
  -> authenticate
& authenticate

//Interpretation
Attacking an asset with multi-factor authentication requires the attacker to obtain both the password and the additional code, e.g., produced by a separate authenticator or sent via text messages to a user.








Defense (#)

// Simplified syntax
# step1
 -> step2

// Intuition
If "step1" is done (enabled, implemented; imp(step1) = 1), then "step2" is not reachable by the attacker

// Example
# encrypted
 -> read

// Interpretation
If data is encrypted, then it cannot be read.








Existence (E)

//Simplified syntax
E existenceCheck
 <- preconditions
 -> step1

// Intuition
If some preconditions for the existenceCheck are satisfied, then "step1" becomes reachable.

// Example
E hasCamera
 <- hardware[Camera]
 -> hijackCamera

// Interpretation
If the system, e.g. a laptop, has a built-in camera, then the camera asset be attacked.








Non-existence (!E)

// Simplified syntax
!E nonExistenceCheck
 <- preconditions
 -> step1

// Intuition
If any preconditions for the nonExistenceCheck are NOT satisfied, then "step1" becomes reachable.

// Example:
!E noAntiMalwareInstalled
 <- applications[AntiMalware]
 -> infectWithMalware

// Interpretation:
If anti-malware software not present on a host system, then it can be infected with malware.








Require (<-)

// Simplified syntax
E existenceCheck
 <- precondition1,
    precondition2
 -> step1
!E nonExistenceCheck
 <- preconditionA,
    preconditionB
 -> step2

// Intuition
Specifies all preconditions to check for existence and non-existence checks. Either precondition1 or precondition2 must exist to reach step1. Similarly, either preconditionA or preconditionB must NOT exist to reach step2.

// Example:
TODO

// Interpretation:
TODO








TypeOf ([])

// Simplified syntax
| step1
  -> assets[Foo].bar

// Intuition
Designates a sub-class Foo of some associated asset to perform operations on.

// Example:
| attackLinux
  -> computers[LinuxMachine].attack

// Interpretation:
The step attackLinux only affects associated computers of the LinuxMachine sub-class.








Union (\/)

// Simplified syntax
foo \/ bar

// Intuition
Designates the union of the sets of associated assets designated by the roles foo and bar.

// Example:
| attackWindowsOrLinux
  -> computers[WindowsMachine] \/ computers[LinuxMachine]

// Interpretation
The step attackWindowsOrLinux affects all associated Windows and Linux machines.








Intersection (/\)

// Simplified syntax
foo /\ bar

// Intuition
Designates the intersection of the sets of associated assets designated by the roles foo and bar.

// Example
| attackWindowsAndLinux
  -> computers.operatingSystems[Windows] /\ computers.operatingSystems[Linux]

// Interpretation
The step attackWindowsAndLinux affects only the set of associated computers which has both Windows and Linux installed.








Difference (-)

// Simplified syntax
foo - bar

// Intuition
Designates the intersection of the sets of associated assets designated by the roles foo and bar.

// Example
| attackWindowsNotLinux
  -> computers.operatingSystems[Windows] - computers.operatingSystems[Linux]

// Interpretation
The step attackWindowsNotLinux affects only the set of associated computers which has Windows, but not Linux, installed.








Collect operator (X.A)

// Simplified syntax
foo.bar

// Intuition
Designates the all associated assets foo, then, the assets or elements bar.

// Example
Asset Computer
{
  operatingSystems.hack
}
// Interpretation
Refers to the attack step 'hack' for all operating systems associated with Computer assets.








Transitive operator (X*.A)

// Simplified syntax
foo*.bar

// Intuition
Recursively follows foo until the end is reached, then, performs bar.

// Example
Asset Computer
{
  folders.subFolder*.access
}
// Interpretation
Refers to all folders associated with computer assets. Recursively follows all subFolder associations as far as possible, then, performs access.








Let substitution (X().A)

// Simplified syntax
let foo = bar,
foo().step

// Intuition
Substitutes foo() with the contents of the corresponding let expression, i.e., foo().step = bar.step.

// Example
Asset Computer
{
  let allFolders = folders.subFolder*,
  allFolders().access
}
// Interpretation
Refers to all folders associated with computer assets. Recursively follows all subFolder associations as far as possible, then, performs access.








Hidden step

// Simplified syntax
| foo @hidden
  -> bar
| bar

// Intuition
The step foo is functional but will not be visualized by securiCAD. The step bar appears normally.

// Example
Asset Password
{
  | guess
    -> accounts.passwordObtained
}

Asset MFAToken
{
  | steal
    -> accounts.tokenObtained
}

Asset Account
{
  | passwordObtained @hidden
    -> compromise
  | tokenObtained @hidden
    -> compromise
  & compromise
}
// Interpretation
Compromising the Account requires both a password and multi-factor authentication token. Guessing the password, stealing the token, and compromising the account will be reported. However, the "helpers" passwordObtained and tokenObtained are informative end users and are, therefore, not visualized.








Debug step

// Simplified syntax
asset HelloWorld
{
| step 
  -> let someAssets = foo /\ bar,
     someAssets[Foo].test
}

asset Foo
{
| test @debug
}
// Intuition
The step test can be called to control which Foo assets were included after the intersection operation during development. Both someAssets[Foo].test and the test step itself will be removed upon packaging a production version.

// Example
Asset Computer
{
  | findWindowsAndLinux
    -> let allTargets = computers.operatingSystems[Windows] /\ computers.operatingSystems[Linux],
       allTargets[Windows].foundWindows
}

Asset Windows extends OperatingSystem
{
  | foundWindows @debug
}

// Interpretation
The let expression attempts to find all associated computers with both Windows and Linux installed. Before writing more logic, the foundWindows step is used to test if all Windows hosts were found as expected while testing the language. Since foundWindows has no value outside of testing, it can be omitted when shipping for securiCAD.








Trace origin

// Simplified syntax
asset Gamma
{
  & activate @trace
  | calledByAlpha
    -> activate
  | calledByBeta
    -> activate
}

asset Alpha
{
  | goToBeta
    -> let allBeta = allGamma.allBeta,
       let someBeta = myBeta /\ allBeta,
       someBeta.calledByAlpha,
       someBeta.foo
}

asset Beta
  | foo
    -> myGamma.calledByBeta

// Intuition
Alpha can only reach Beta through Gamma. However, the path resulting from someBeta would only display Alpha->Beta. The involved Gamma assets would be omitted, even with the activate step in place. However, by tracing activate, securiCAD will show all relationships between Alpha, Beta, and Gamma.

// Example

asset Policy {
  & satisfy @trace
  | calledByUser
    -> satisfy
  | calledByHost
    -> satisfy
}

asset User {
  let effectivePolicies = policies
  let policyReachableHosts = policies.hosts

  | compromise
    -> effectivePolicies.calledByUser,
       policyReachableHosts.connect
}

asset Host {
  let effectivePolicies = policies
  | connect
    -> effectivePolicies.calledByHost

}
// Interpretation
The security Policy determines which Host machines any User may communicate with. However, policyReachableHosts.connect only provides the path User->Host and omits the offending Policy. Therefore, the satisfy step is traced in the Policy asset.








Risk type

// Simplified syntax
| stepC {C}
| stepCI {C, I}
| stepCIA {C, I, A}

// Intuition
Reaching step C harms confidentiality. Reaching step CI harms confidentiality and integrity. Step CIA harms confidentiality, integrity, and availability.

// Example

asset Data {
  | readOnly {C}
  | readAndAppend {C, I}
  | fullAccess {C, I, A}

// Interpretation
Unauthorized reads harms confidentiality. Being able to append information harms the integrity of the data. Having full access potentially harms all security properties.













            

          

      

      

    

  

    
      
          
            
  
Versioning

This page documents the syntax for the legacy compiler. The documentation for the current foreseeti compiler can be found here [https://github.com/mal-lang/mal-documentation/wiki/MAL-Syntax].




Language Constituents

The following are commonly used keywords in writing MAL specification:


	category Similar to a package in Java. A category consists of one or more assets. The category does not bear semantics, it is only there to enable structure for the language developer.


	asset Similar to a class in Java. An asset could represent physical objects (e.g., a network router) or logical objects (e.g., Credentials, or a Network Service). When the MAL compiler generates the Java code from the MAL specifications, an
asset is translated into a java class.


	abstractAsset Similar to asset, however, it refers to an asset which need not have a class but its properties are inherited by its children. So, in Java generated code, an abstract asset will not have a class, and its properties are included in each child asset.


	extends Similar to inheritance in Java. A extended asset inherits attack steps and defense steps from its parent asset. Also the asset associations are inherited by its extended assets (e.g. we don’t need to create the association for an extended asset X1 to another asset X2, if its parent asset X is already associated with X2.).


	info This keyword is used in MAL to provide information for users (not MAL writers) about the asset and/or the associated attack step and defense mechanism.


	rationale This keyword is used by developers of MAL specification. It tells MAL writers why the attack step is used and other related information.


	association Association is similar to UML association. It is a connection between two assets, and the name of the association is written near the asset. The multiplicity of a composite association can be [1..1], [1..] or [..*] on the composite end.







Basic MAL Example

MAL, like other languages, has its own syntax and alphabet of symbols. Each asset may have attack or defense steps, and can have associations between each other. The MAL symbols can be seen below. The  associations between assets follow the structure of UML class diagrams, where associations are bound to types with cardinalities.

category User {
    asset UserAccount {
        | impersonate
            -> compromise
        | compromise 
            -> computer.stealSecret
    }
}
category Hardware {
    asset Computer {
        & compromise
            -> communicate,
                folder.subFolder+.accessFolder
        | communicate
            -> connect
        | authenticate
            -> compromise
        | vulnerability
            -> compromise
        & access
            -> compromise
        | connect
            -> access
        | stealSecret
            -> compromise
        | firewallBypass
            -> firewall.bypass
        E firewall
            <- firewall
            -> access
    }
    asset Firewall {
        & bypass [probability]
            -> computer.access
        # protected
            -> bypass
    }
    asset Folder {
        | accessFolder
            -> findCredentials
        | findCredentials
    }
}

associations {
    Computer [studentComputer] 1-* <-- Use --> 1 [student] UserAccount
    Computer [teacherComputer] 1 <-- Use --> 1 [teacher] UserAccount
    Computer [computer] 1-* <-- Use --> 1-* [person] UserAccount
    Computer [computer] * <-- Protect --> 1 [firewall] Firewall
    Computer [computer] * <-- Contains --> * [folder] Folder
    Folder [folder] 1 <-- Contains --> * [subFolder] Folder
}








MAL Symbols

| Symbol | Meaning | Description |
| — | — | — |
| –> | Leads to | The successful compromise of this attack step allows the attacker to consequently compromise further attack steps. |
| | | OR | An OR attack step A can be reached if any of the attack steps which refers to A is reached.
| & | AND | An AND attack step A can be reached only if all of the attack steps which refer to A are reached.
| # | Defense | As opposed to attack steps defences are boolean. A defense step represent the countermeasure of an attack step (e.g., passwordBruteForce attack step can be defended by twoFactorAuthentications defense step). While declaring an asset, it is possible to either enable or disable a defense step by setting the defense step to either TRUE or FALSE.
| E | Existence | This symbol is related to $<$– operator. It is used when the existence of a connected/associated asset must be checked. It acts like a defense but the boolean value is automatically assigned based on the existence of the asset. The example can be seen in Appendix, the firewallProtected attack step under \texttt{Computer} asset.
| 3 | Non-existence | It is the same as above but this time a check for non-existence is being made (i.e. when the associated asset does not exist, the boolean value will be true).
| [ ]  | TypeOf | Reaching an attack step that traverses a set of assets can now be further specified by child asset if a parent asset has been extended. For example, if Dataflow has children assets Inbound and Outbound, we specify as dataflows[Outbound]
| \/ | Union | Union operations of same-typed sets.
|  /\ | Intersection | Intersection operations on same-typed sets.
| X.A | Collect operator | Attack step A of the asset X is referenced. Other asset X is referenced to reach the attack step A.
| X+.A | Transitive operator | It can be used to refer many attack steps without writing all of them.
|  +> | Add operator | This happens to the child asset. One attack step (e.g. access) under the child asset leads to its specific attack steps besides its corresponding attack step of the parent asset.
| TTC | Time-To-Compromise | TTC is a probability distribution reflects an attacker’s ability of compromising an asset. Attack steps can have a TTC to execute. the MAL simulations calculate/aggregate TTC for models/scenarios.




Probability Distributions

Some attack steps have probability distributions to express their uncertainties. MAL has four probability distributions to choose from:


	ExponentialDistribution


	UniformDistribution


	GammaDistribution


	BernoulliDistribution







MAL Coding Standard

MAL specification follows a coding standard similar to Java. Please consider the following coding standard when creating your MAL language.


	Asset names start with a capital letter


	Attack steps are following camelCase


	Role names are lower case


	Comments in MAL files can be introduced by “//”


	It is recommended that asset names are not repeated in the attack step








            

          

      

      

    

  

    
      
          
            
  This is the developers’ guide to MAL. For technical details, see the MAL compiler wiki [https://github.com/mal-lang/malcompiler/wiki/].


Versioning

This page documents the latest changes based on the foreseeti MAL compiler. The syntax for the legacy compiler can be found here [https://github.com/mal-lang/mal-documentation/wiki/MAL-Syntax-(legacy)].




Language Constituents

The following are commonly used keywords in writing MAL specification:


	define [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#define] Defines information for entire MAL projects. The syntax is ‘#key: “value”’, for example ‘#version: “1.0.0”. The keys #id and #version must be present in every project. For more information, see the MAL compiler wiki [https://github.com/mal-lang/malcompiler/wiki/MAL-language-specification#define].


	include [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#include] “path/to/specification.mal” Imports another MAL specification. Enables references to assets and elements defined in separate specifications. Paths are relative to the specification with the include statement.


	category [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#category] Similar to a package in Java. A category consists of one or more assets. The category does not bear semantics, it is only there to enable structure for the language developer.


	asset [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#asset] Similar to a class in Java. An asset could represent physical objects (e.g., a network router) or logical objects (e.g., Credentials, or a Network Service). When the MAL compiler generates the Java code from the MAL specifications, an
asset is translated into a java class.


	extends [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#extends] Similar to inheritance in Java. A child asset inherits attack steps and defense mechanisms from its parent. Child steps and mechanisms override their parent counterparts by default if the same names are used. Let expressions and associations are also inherited by extended assets (e.g. we don’t need to create the association for an extended asset Linux to another asset Application, if its parent asset Machine is already associated with Application.).


	abstract asset [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#abstract-asset] Similar to asset except it cannot be instantiated, only extended. An abstract asset need not have a class but its properties are inherited by its children. So, in Java generated code, an abstract asset will not have a class, and its properties are included in each child asset.


	user info [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#user-info] This keyword is used in MAL to provide information for end users about the asset and/or the associated attack step and defense mechanism.


	developer info [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#developer-info] This keyword is used by developers of MAL specifications. It tells other MAL writers why the attack step is used and other related information.


	modeler info [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#modeler-info] This keyword is used to provide information to modellers or parser developers. It can be used to communicate assumptions or parsing requirements, which might otherwise be ambiguous.


	let [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#let] This keyword is used to associate a given expression with a specific reusable name. The syntax is ‘let <nameHere> = <MAL expression>’. Let expressions can be defined within the scope of entire assets, specific attack steps, and defense mechanisms. Let expressions can be referred to across separate assets like attack steps, assuming appropriate associations exist.


	association [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#association] Association is analogous to UML class diagram associations. It denotes a potential bidirectional structural relationship (physical or logical) between two assets. The syntax is ‘Asset1 [role1] <multiplicity1> <– AssociationName –> <multiplicity2> [role2] Asset2’. The multiplicity of a composite association can be [1], [1..*] or [*] on the composite end. Role names determine the directional references to other assets used in code. For example, Asset1 references Asset2 as role2.A, where A is an attack step. Conversely, Asset2 references Asset1 as role1.A.







MAL Symbols

| Symbol | Meaning | Description |
| — | — | — |
| -> | Leads to [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#leads-to--] | The successful compromise of this attack step allows the attacker to consequently compromise further attack steps. Also specifies steps affected by defenses and existence checks. |
|  +> | Append [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#append-] | Child assets only. When parent and child assets have overlapping elements, e.g. | access, the expressions defined for the child access are appended to those of the parent access. The child expressions will otherwise override those of the parent.
| | | OR [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#or-step-] | An OR attack step A can be reached if any of the attack steps which refers to A is reached.
| & | AND [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#and-step-] | An AND attack step A can be reached only if all of the attack steps which refer to A are reached.
| # | Defense [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#defense-] | As opposed to attack steps defenses are boolean. A defense step represents the countermeasure of an attack step (e.g., passwordBruteForce attack step can be defended by twoFactorAuthentications defense step). While declaring an asset, it is possible to either enable or disable a defense step by setting the defense step to either TRUE or FALSE. It is also possible to assign a Bernoulli distribution to govern the activation.
| E | Existence [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#existence-e] | It is used when the existence of connected/associated assets, given by <-, must be checked. The specified attack steps are reachable when at least one designated asset exist.
| !E | Non-existence [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#non-existence-e] | Same as Existence, except the specified attack steps are reachable when at least one designated asset does NOT exist.
| <- | Require [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#require--] | Denotes which associated assets are required by the current expression. Used to specify preconditions for Existence and Non-existence. |
| [ ]  | TypeOf [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#typeof-] | Reaching an attack step that traverses a set of assets can now be further specified by child asset if a parent asset has been extended. For example, if Dataflow has children assets Inbound and Outbound, we specify as dataflows[Outbound]
| \/ | Union [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#union-] | Union operations of same-typed sets. (Backslash + forward slash)
|  /\ | Intersection [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#intersection-] | Intersection operations on same-typed sets. (Forward slash + backslash)
| - | Difference [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#difference--] | Difference operations on same-typed sets.
| X.A | Collect operator [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#collect-operator-xa] | Attack step A of the asset X is referenced. Other asset X is referenced to reach the attack step A.
| X*.A | Transitive operator [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#transitive-operator-xa] | Recursively follow X until the end is reached, then, perform A. For example, folder.subFolder.subFolder.subFolder.access = folder.subFolder*.access. Hanging transition are allowed for let expression, e.g., let allFolders = folders.subFolder*.
| X().A | Let substitution [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#let-substitution-xa] | The parentheses after X denote that it refers to the let expression labeled ‘X’. A is performed on the assets given by X.
| @hidden | Hidden step [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#hidden-step] | (securiCAD only) Attack steps tagged as hidden remain functional but will not be shown in attack paths.
| @debug | Debug step [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#debug-step] | (securiCAD only) Attack steps tagged as debug exist while testing but are removed from the language while exporting for securiCAD.
| @trace | Trace origin [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#trace-origin] | (securiCAD only) Attack steps tagged for tracing force securiCAD into reporting their root cause. The typical use case is highlighting which intermediary assets were involved in a set expression evaluation. For example, which intermediary security policy assets were reached between userAccount.assume->action.perform?
| AS {C, I, A} | Risk type [https://github.com/mal-lang/mal-documentation/wiki/MAL-Code-Examples#risk-type] | (securiCAD only) Attack steps (AS) can have specific types of risks associated with them. It can be any combination of Confidentiality (C), Integrity (I), and Availability (A). Risk types are used by securiCAD together with probability distributions to calculate the total risk.
| TTC | Time-To-Compromise | TTC is a probability distribution reflects an attacker’s ability of compromising an asset. Attack steps can have a TTC to execute. the MAL simulations calculate/aggregate TTC for models/scenarios.




Probability Distributions

For detailed information regarding distributions, see Supported distribution functions (MAL compiler) [https://github.com/mal-lang/malcompiler/wiki/Supported-distribution-functions].

Attack steps may have probability distributions set to express their uncertainties. MAL has the following to choose from:


	Bernoulli(p)


	Binomial(n, p)


	Exponential(λ)


	Gamma(k, θ)


	LogNormal(μ, σ)


	Pareto(x, α)


	TruncatedNormal(μ, σ^2)


	Uniform(a, b)







MAL Coding Standard

MAL specification follows a coding standard similar to Java. Please consider the following coding standard when creating your MAL language.


	Asset names are PascalCased.


	Attack steps, defense mechanims, and role names are camelCased.


	The attack step definition syntax is ‘<\type> nameHere @tag [ProbabilityDistribution(params)]’. Tags and probability distributions are optional.


	Multiple expressions can be written under both -> and <- operators. They should be delimited by a comma and line break. Furthermore, the first expression should be on the same line as the operator itself.


	The defense mechanism definition syntax is ‘# nameHere @tag [Bernoulli(p)]’. The tags and Bernoulli distribution are optional.


	Comments in MAL files can be introduced by “//”


	It is recommended that asset names are not repeated in the attack step


	Associations are defined at the end of each MAL specification.


	Inline documentation using the three info elements is preferred over other formats.








            

          

      

      

    

  

    
      
          
            
  The method of how to construct a domain specific MAL begins with a clear understanding of the system that one wants to model. It is important to spend time learning about the system, because being comfortable in the domain will make modeling of the language much easier. The domain can often be studied through research paper and literature.  At other times it is useful to also consult experts and conduct interviews to ask any additional questions.

After enough knowledge about the system is learnt, it is time to decide what the scope of the MAL should be. Modeling a system can easily become too complex if the scope is set too broad or too detailed. It is recommended to start with a smaller scope from the beginning and then broaden the scope as required with time.

Once the domain is well known and the scope is set, it is time to create the language. Begin by defining all the possible attack steps. After the attack steps are known, it is possible to define all the defences for these steps as well as their relationships.

It is important to thoroughly analyze each attack step in detail and find any potential defences as well as resulting new attack steps. A successful attack step may result in another attack step.  Also, there are situations where combined attack steps are required for a resulting attack step to be reached. All these relationships must be found and stated clearly.

With the MAL specification completed, it is time to compile and validate the language.



            

          

      

      

    

  

    
      
          
            
  
Welcome to the malcompiler wiki!





            

          

      

      

    

  

    
      
          
            
  
Grammar

The grammar of MAL can be found at https://github.com/mal-lang/malcompiler/blob/master/grammar. The grammar is not written to be machine readable, but instead to be human readable yet correct.




Details

A MAL specification is a text file consisting of define, include, category, and associations declarations. These declarations can appear in the file in any order and in any number.


Comments

Comments can appear anywhere in the MAL specification file. There are two types of comments, line comments and block comments.

A line comment starts with // and ends at the end of that line. For example:

#id: "com.example.lang" // This is the unique identifier of the language
#version: "1.2.3"       // This is the version of the language





A block comment starts with /* and ends with */, and do not nest. For example:

/**
 * These are the required project coordinates of the language
 */
#id: "com.example.lang"
#version: "1.2.3"





Comments should not be used for documentation of the language. For documentation, use meta declarations instead.




Define

define declarations look like this:

#key: "value"





For example:

#id: "com.example.lang"
#version: "1.2.3"
#foo: "bar"





Every key must be unique across all MAL specifications in a project, and the keys #id and #version must be defined in every MAL project.

#id can be defined as any string, but by convention it should use reverse domain name notation [https://en.wikipedia.org/wiki/Reverse_domain_name_notation] to uniquely identify the MAL project.

#version must be defined as a sequence of three numbers separated by dots. The convention is to use semantic versioning [https://semver.org/] without additional labels.




Include

include declarations look like this:

include "file"





For example:

include "other1.mal"
include "subdir/other2.mal"
include "../other3.mal"





Including an other MAL specification is equivalent to replacing the include declaration with the contents of the included file. If the path of the include declaration is relative, it will be evaluated as relative to the MAL specification containing the include declaration.




Category

category declarations look like this:

category <id>
  <meta>*
{
  <asset>*
}





A category is a group for assets. The category name does not have to be unique to a category declaration.




Asset

TODO




Associations

associations declarations look like this:

associations {
  <association>*
}





See association for more details.




Association

association declarations look like this:

<id> [<id>] <mult> <-- <id> --> <mult> [<id>] <id>
  <meta>*





For example:

Network [networks] * <-- NetworkAccess --> * [hosts] Host





These declarations define an association between two assets. In the example above, the association is called NetworkAccess. A Host can be connected to zero or more Networks through the field networks and a Network can be connected to zero or more Hosts through the field hosts.

The identifiers in brackets, [networks] and [hosts] above, are called fields. These can be used in attack steps to specify what can be reached. For example:

category Example {
  asset A {
    | compromise
      -> b.compromise
  }
  asset B {
    | compromise
  }
}
associations {
  A [a] * <-- L --> * [b] B
}





Because an association is defined between assets A and B, the field b can be used in the attack step compromise in the asset A to specify that if an asset A is compromised, all associated assets B are also compromised.

The identifiers between arrows, e.g. <-- NetworkAccess -->, are called link names. These names describe the association symmetrically, and gives each association a proper name.

Multiplicities in association specify how the association shall be used in models. The following table describes all valid multiplicities in associations:

| Association                    | Meaning                                                                     |
| —————————— | ————————————————————————— |
| A [a] * <-- L --> 1 [b] B    | Asset A must be connected to exactly one asset B through the field b  |
| A [a] * <-- L --> * [b] B    | Asset A can be connected to zero or more assets B through the field b |
| A [a] * <-- L --> 0..1 [b] B | Asset A can be connected to zero or one asset B through the field b   |
| A [a] * <-- L --> 0..* [b] B | Same as A [a] * <-- L --> * [b] B                                         |
| A [a] * <-- L --> 1..1 [b] B | Same as A [a] * <-- L --> 1 [b] B                                         |
| A [a] * <-- L --> 1..* [b] B | Asset A must be connected to one or more assets B through the field b |

In all associations in the table above, the asset B can be connected to zero or more assets A through the field a.




ID

An id must start with an upper case or lower case letter or an underscore ([A-Za-z_]) and can be followed by zero or more letters, underscores, or digits ([A-Za-z_0-9]*). Examples: Host, networkAccess, network_access_1

By convention, categories and assets should start with an upper case letter. Attack steps, variables, fields, and defines should start with a lower case letter. All identifiers should use camel case [https://en.wikipedia.org/wiki/Camel_case] and underscores are discouraged.




Meta

meta declarations attach different types of information to category, asset, attackstep, and association declarations.

Valid meta declarations are:

user info: "Information for users"
developer info: "Information for language developers"
modeler info: "Information for modelers"





These declarations should be used to document the language.

Example:

category Example
  user info: "The 'Example' category contains example assets"
{
  asset Asset1
    modeler info: "Connect the attacker to this asset"
  {
    | attack
      modeler info: "This should be the entry point for the attacker"
      developer info: "This attack step is used to reach compromise on all sub assets"
      -> subAsset*.compromise
    | compromise
      user info: "The attacker has full access on this asset"
  }
}

associations {
  Asset1 [superAsset] 0..1 <-- Hierarchy --> 0..1 [subAsset] Asset1
    user info: "Assets can be connected in a hierarchy"
}











            

          

      

      

    

  

    
      
          
            
  
Attack steps

Attack steps of type AND, & and OR, | use probability distributions to specify the TTC (time to compromise) of that attack step. For example:

category Systems
{
  asset Computer
  {
    | compromise [Exponential(0.1)]
  }
}





This means that the time to perform the attack step compromise on an asset Computer is expressed with an exponential distribution with a rate of 0.1 (with a mean value of 10 days).

The following distributions are supported:

| Distribution | Parameters | Limits | Expected value |
| ———— | ———- | —— | ————– |
| Bernoulli | p, probability | 0 <= p <= 1 | E[Bernoulli(p)] = p |
| Binomial | n, trials and p, probability | 0 <= n, 0 <= p <= 1 | E[Binomial(n, p)] = n * p |
| Exponential | λ, rate | 0 < λ | E[Exponential(λ)] = 1 / λ |
| Gamma | k, shape and θ, scale | 0 < k, 0 < θ | E[Gamma(k, θ)] = k * θ |
| LogNormal | μ, mean and σ, standard deviation | 0 < σ | E[LogNormal(μ, σ)] = e^(μ + σ^2 / 2) |
| Pareto | m, minimum and α, shape | 0 < m, 0 < α | E[Pareto(m, α)] = infinity if m <= 1, otherwise (m * α) / (α - 1) |
| TruncatedNormal | μ, mean and σ, standard deviation | 0 < σ | E[TruncatedNormal(μ, σ)] = μ |
| Uniform | min and max | min <= max | E[Uniform(min, max)] = (min + max) / 2 |

Distributions can also be combined with addition, +, subtraction, -, multiplication, *, division, /, exponentiation ^, and parentheses (, ).

The following ordinal distributions are supported:

| Ordinal distribution | Definition |
| ——————– | ———- |
| EasyAndCertain | Exponential(1) |
| EasyAndUncertian | Bernoulli(0.5) |
| HardAndCertain | Exponential(0.1) |
| HardAndUncertain | Bernoulli(0.5) * Exponential(0.1) |
| VeryHardAndCertain | Exponential(0.01) |
| VeryHardAndUncertain | Bernoulli(0.5) * Exponential(0.01) |

The following constants are supported:

| Constant |
| ——– |
| Infinity |
| Zero |


Bernoulli behaviour

An attack step with a TTC function Bernoulli(p) means that the attack step can be performed immediately with a probability of p, and not at all with a probability of 1 - p.

If a Bernoulli distribution is used in multiplication, e.g. Exponential(λ) * Bernoulli(p), that means that the TTC of the attack step is Exponential(λ) with a probability of p, and Infinity with a probability of 1 - p.

Bernoulli distributions can also be used in addition, but not in subtraction, division, or exponentiation.






Defenses

Defenses, #, are either enabled or disabled with a probability. For example:

category Systems
{
  asset Computer
  {
    # protected [Bernoulli(0.4)]
  }
}





This means that the probability that the defense protected is enabled for an asset Computer is 0.4 (40%).

Defenses can use the following probabilies:

| Probability | Explanation |
| ———– | ———– |
| Bernoulli(p) | Defense is enabled with probability p |
| Enabled | Same as Bernoulli(1) |
| Disabled | Same as Bernoulli(0) |

